Szélsőérték

Dr. Sárvári Csaba

Mérnöki Matematika Tanszék

Pollack Mihály Műszaki Kar

Pécsi Tudományegyetem

Email: sarvari@witch.pmmf.hu
E-Learning: http://www.matserv.pmmf.hu/e-learning/

Ebben a fejezetben megismerkedünk azokkal a tételekkel, amelyek segítségével megvizsgálhatjuk a függvény monotonitását és szélsõértékét. A helyi szélsőérték létezésére vonatkozó szükséges feltételt mutatunk be, majd a monotonitás vizsgálatára vonatkozó és a helyi szélsőérték létezését biztosító, más szóval a helyi szélsõérték létezésére vonatkozó elégséges feltételeket tárgyalunk. Eljárásokat adunk a szélsõérték kiszámítására.

A differenciálszámítás alaptételei

A $f'(x)$ (másként jelölve: $\frac{d}{dx} f(x)$, még másként $D(f)(x)$) differenciálhányados segítségével az $f(x)$ függvény viselkedését jellemezhetjük.

Tétel

Ha az f függvény az x_0 pontban differenciálható és $f'(x_0)>0$ ($f'(x_0)<0$), akkor az x_0-hoz elég közel fekvõ $x < x_0$ értékekre $f(x) < f(x_0)$ ($f(x_0) < f(x)$) ill. $x_0 < x$ esetén $f(x_0) < f(x)$ ($f(x) < f(x_0)$).
Szavakban kifejezve: Ha a derivált az x_0 pontban pozitív, akkor f az x_0 ponton szigorúan monoton növekvőleg halad át, ha a derivált az x_0 pontban negatív, akkor szigorúan monoton csökkenőleg halad át.

Bizonyítás.

Tekintsük az $f'(x_0)>0$ esetét.Tudjuk, hogy

$$\lim_{x \to x_0} \left(\frac{f(x) - f(x_0)}{x - x_0} \right) = f'(x_0)$$

Használjuk föl a következő segédtételt:
Ha az f függvénynek az x_0 pontban pozitív határértéke van, akkor az x_0 egy környezetében $0 < f(x)$.
Ha mármost $f'(x_0)>0$, akkor az x_0-hoz elég közel fekvõ x értékekre

$$0 < \frac{f(x) - f(x_0)}{x - x_0} ,$$

ez pedig azt jelenti, hogy ha $x < x_0$, akkor $f(x) < f(x_0)$, s ha $x_0 < x$, akkor $f(x_0) < f(x)$, vagyis f az x_0 ponton növekvõleg halad át.
A tétel "megfordítása": Ha f az x_0 pontban differenciálható és az x_0-on növekvõleg (csökkenőleg)

halad át, akkor f '(x_0)≥0
(f '(x_0)≤0).

Alapvető jelentőségű a következő tétel, amely a differenciálható függvény helyi szélsőértékének létezésére vonatkozó szükséges feltételt jelent.

Tétel. (Fermat tétele)

Legyen az f függvény az X intervallumban értelmezve és vegye föl az intervallum belső c pontjában a helyi szélsőértékét (maximumát vagy minimumát). Ha f a c pontban differenciálható, akkor

$$\left.\frac{d}{dx} f\right|_{x=c} = \mathbf{0}$$, másképp jelölve: **(f ' (c) =0).**

Másképpen megfogalmazva: a c pontban differenciálható f függvénynek az $x = c$ helyen csak akkor **lehet** helyi szélsőértéke,

ha f ' (c) =0.

Bizonyítás.

Vegye föl az f függvény az x_0 pontban a helyi szélsőértékét. Ha f '(x_0) nullától különböző volna, akkor ellentondásra jutnánk, hisz ekkor vagy növekvõleg, vagy csökkenőleg kellene f-nek az x_0-on áthaladnia.

A következõkben a differenciálszámítás középérték tételeit tárgyaljuk:

Rolle tétele.

Ha az f függvény az $[a, b]$ zárt intervallumban folytonos, az (a,b) nyílt intervallumon differenciálható és $f(a) = f(b)$, akkor
van az (a,b) nyílt intervallumban olyan ξ érték, amelyre

$$\left.\left(\frac{d}{dx} f(x)\right)\right|_{x=\xi} = 0 \qquad (f'(\xi) = 0)$$

A tétel geometriailag megfogalmazva azt jelenti, hogy az (a,b) nyílt intervallumban van olyan ξ hely, amelyben a függvény érintője párhuzamos az x tengellyel.
Úgy is fogalmazhatunk, hogy ha a differenciálható f függvény adott $[a, b]$ intervallumra eső átlagos változási sebessége, azaz a különbségi hányadosa 0 értékű, akkor van az intervallum belsejében olyan pont, ahol a derivált értéke 0.
Szemléltessük mindezt az $f(x) = \sin(x) + \frac{\sin(2x)}{2}$ függvény példáján! Az intervallum legyen a $[0, \pi]$. Felvesszük a függvényt és rögtön ábrázoljuk is.

```
> restart:
> f:=x->sin(x)+1/2*sin(2*x);
```

$$f := x \rightarrow \sin(x) + \frac{1}{2}\sin(2\,x) \tag{1.1}$$

```
> plot(f(x), x=0..2*Pi);
```



Elõször azt rögzítjük, hogy a Rolle tétel feltételei teljesülnek. Valóban a $f(x) = \sin(x) + \frac{\sin(2\,x)}{2}$ a $[0, \pi]$ *zárt intervallumon folytonos* (valójában az egész számegyenesen az), és *deriválható a (0, π) nyitott intervallumon* (valójában mindenhol differenciálható). Végül a Rolle tétel harmadik feltétele, hogy az intervallum végpontjaiban felvett függvényértékeknek egyenlõnek kell lennie. Ez is teljesül, hiszen $f(0) = f(\pi) = 0$.
A Rolle tétel feltételei tehát teljesülnek, így annak konklúziója is, ami pedig azt biztosítja, hogy a deriváltnak (0, π) nyitott intervallumban létezik zérushelye. Az alábbi utasítás outputja összehangban van az itt elmondottakkal, hiszen ...

```
> zh:=[solve(D(f)(x))];
```

$$zh := \left[\frac{1}{3}\pi, \pi\right] \tag{1.2}$$

```
>
```

... a $\frac{\pi}{3}$ zérushelye a D(f) derivált függvénynek, és benne van a (0, π) nyitott intervallumban.

Ábrázoljuk a függvényt és a $\frac{\pi}{3}$ abszcisszájú, az x tengellyel párhuzamos érintõjét

```
>
```

A Rolle- tétel általánosítása a következő tétel:

Tétel. (Lagrange tétele)

Ha $f(x)$ az $[a, b]$ zárt intervallumon folytonos és
az (a,b) nyílt intervallumon differenciálható,

akkor az (a,b) intervallumban van olyan ξ amelyre

$$f'(\xi) = \frac{f(b) - f(a)}{b - a}.$$

Másképp jelölve

$$\left(\frac{d}{dx} f(x)\right)\bigg|_{x = \xi} = \frac{f(b) - f(a)}{b - a}$$

A tétel szerint a feltételeket teljesítő f függvény esetén van az (a,b) intervallumban olyan ξ hely, ahol a függvény változási sebessége egyenlõ az $[a, b]$ intervallumbeli átlagos változási sebességgel. Geometriailag ez azt jelenti, hogy az adott pontbeli érintõ párhuzamos az intervallum végpontjaihoz tartozó szelővel.
Az alábbi Langrange eljárás a Langrange tételt állítását szemlélteti. A feltételek teljesülése esetén ábrázolja a szelõt és a vele párhuzamos érintõt, egyébként pedig hibaüzenetet küld.

```
> Lagrange:=proc(f,a,b)
  local  abra1,abra2,  vonal1,  vonal2,  xi,  szelo:
  if  not(iscont(f(x),x=a..b,'closed'))  then
   ERROR(`Nem folytonos az [a,b] zart intervallumon`):
  fi:
  if  not(iscont(D(f)(x),x=a..b))  then
   ERROR(`Nem derivalhato az (a,b)  nyilt intervallumon`);
  fi:
  xi:=fsolve(D(f)(x)=(f(b)-f(a))/(b-a),x=a..b);
  abra1:=plot(f(x),x=a-0.1..b+0.1,color=blue):
  abra2:=plot(D(f)(xi)*(x-xi)+f(xi),x=xi-2..xi+2,color=red,
  thickness=2):
  vonal1:=plottools[line]([a,0],[a,f(a)],color=green,linestyle=
  3):
  vonal2:=plottools[line]([b,0],[b,f(b)],color=green,linestyle=
  3):
  szelo:=plot((f(b)-f(a))/(b-a)*(x-a)+f(a),x=a..b,color=
  magenta):
  plots[display]([vonal1,vonal2,szelo,abra1,abra2])
  end:
```

Hívjuk meg a Lagrenge eljárást a sin függvénnyel, a vizsgált intervallum legyen [-4,3] .
Megjegyezzük, hogy a $f(x) = \sin(x)$ eljesíti a Lagrange tétel feltételeit, hiszen a színusz függvény a valós számos halmazán deriválható és így ott folytonos is.
Javasoljuk, hogy próbálja ki más intervallumokon a Langrange eljárást. Ehhez csak az alábbi utasításban szereplõ intervallum végpontokat kell megvaltoztatnia és újra végrehajtani az utasítást (értsd. leütni az Enter billentyût).

```
> Lagrange(sin,-4,3);
```



Az eljárás szakaszonként adott (piecewise) függvényekre is alkalmazható:

```
> g:=x->piecewise(x<=0,-x*sin(x),x*sin(x));
```

$$g := x \rightarrow piecewise(x \leq 0, -x\sin(x), x\sin(x)) \tag{1.3}$$

```
> Lagrange(g,-2,3);
```



Tekintsünk egy példát arra az esetre, amikor a feltételek valamelyike nem teljesül:

Példa

Vizsgáljuk meg, hogy teljesülnek-e a Lagrange-tétel feltételei az

$$f(x) = \begin{cases} -x^2 + 8\,x - 14 & x \le 5 \\ -\dfrac{1\,x}{2} + \dfrac{7}{2} & 5 < x \end{cases}$$

függvényre!

Megoldás

Vegyük föl a szakaszonként adott függvényt!

```
> f:=x->piecewise(x<=5, -x^2+8*x-14, x>5, x^2-7*x+11);
```

$$f := x \to piecewise\left(x \le 5, -x^2 + 8\,x - 14, 5 < x, x^2 - 7\,x + 11\right) \tag{1.4}$$

```
> 'f(x)'=f(x);
```

$$f(x) = \begin{cases} -x^2 + 8\,x - 14 & x \le 5 \\ x^2 - 7\,x + 11 & 5 < x \end{cases} \tag{1.5}$$

Ábrázoljuk a függvényt és deriváltját közös koordinátarendszerben:

```
> plot([f(x),D(f)(x)],x=4..6,discont=true,color=[blue,red],
  thickness=2,legend=["f(x)","D(f)(x)"]);
```



A függvény folytonos az [5, 7] intervallumon, az $x = 5$ helyen azonban nem differenciálható.

```
> iscont(f(x),x=4..6);
```

$$true \quad (1.6)$$

```
> iscont(D(f)(x),x=4..6);
```

$$false \quad (1.7)$$

```
> Lagrange(f,4,6);
Error, (in Lagrange) Nem derivalhato az (a,b) nyilt intervallumon
```

A következõ utasítássorozat különbözõ intervallumok egymásutánján **animáció**val szemlélteti a Langrange tétel állítását.

```
> f:=x->sin(x)+cos(2*x);
```

$$f := x \rightarrow \sin(x) + \cos(2\,x) \quad (1.8)$$

```
> p:=seq(Lagrange(f,-4,4.6-k*0.2),k=0..12):
> plots[display]([p],insequence=true);
```



>

Az intervallumbeli monotonitás és a derivált

Tétel.

Legyen az $f(x)$ függvény az X intervallumon (az X intervallum lehet nyílt, zárt, félig nyitott, véges vagy végtelen) definiált és folytonos és az intervallum belsejében differenciálható függvény. Ekkor f az X intervallumon akkor és csak akkor monoton növekvő (monoton csökkenő), ha

$$0 \le f'(x) \quad (f'(x) \le 0)$$

feltétel teljesül minden az X intervallum belsejéből vett x értékre.

Bizonyítás.

A bizonyítást az első esetre végezzük el. Először megmutatjuk, hogy **a feltétel szükséges**.
Tegyük föl, hogy az f az X intervallumon monoton növekvő. Ekkor tetszőleges x_0 értéket választva az intervallum belsejéből, minden $x \in X$ esetén

$$0 \le \frac{f(x) - f(x_0)}{x - x_0}$$

teljesül. De ekkor

$$0 \le \lim_{x \to x_0} \left(\frac{f(x) - f(x_0)}{x - x_0} \right)$$

hiszen, ha egy függvény az x_0 egy környezetében nem negatív- s különbségi hányadosról épp ezt tettük föl- akkor ott nem lehet negatív határértéke.
Most megmutatjuk, hogy **a feltétel elégséges.** Teljesüljön $0 \le f'(x)$ minden x-re, amely az X intervallum belsejében van.Válasszunk két tetszőleges értéket az intervallum belsejéből, legyenek ezek x_1 és x_2 $(x_1 < x_2)$. A Lagrange-féle középérték tétel alapján:

$$f(x_2) - f(x_1) = f'(c) \cdot (x_2 - x_1),$$

ahol $x_1 < c$ és $c < x_2$.A derivált értéke az intervallum minden belső pontjában nem- negatív, így

$$0 \le f'(c)$$

miatt

$$f(x_1) \le f(x_2),$$

tehát f valóban monoton növekvõ (nem csökkenõ).

Megjegyzés.

A bizonyítás második részéből kiolvasható, hogy ha $0 < f'(x)$ $(f'(x) < 0)$ minden az X intervallum belsejéből vett x értékre, akkor az f függvény szigorúan monoton növekvõ (szigorúan monoton csökkenõ) az (a,b) intervallumban.

Példa.

Határozzuk meg az $f(x) = x^3 - 3x^2 + 2x$ függvény monotonitási intervallumait!

Megoldás.

```
> f:=x->x^3-3*x^2+2*x;
```

$$f := x \to x^3 - 3x^2 + 2x \tag{2.1}$$

Határozzuk meg a deriváltat:

```
> df:=unapply(diff(f(x),x),x);
```

$$df := x \to 3x^2 - 6x + 2 \tag{2.2}$$

Keressük meg a derivált zérushelyeit:

```
> zh:=solve(df(x));
```

$$zh := 1 + \frac{1}{3}\sqrt{3}, 1 - \frac{1}{3}\sqrt{3} \tag{2.3}$$

Megkereshetjük azokat az intervallumokat is, ahol a derivált pozitív:

```
> solve(diff(f(x),x)>0);
```

$$RealRange\left(-\infty, Open\left(1 - \frac{1}{3}\sqrt{3}\right)\right), RealRange\left(Open\left(1 + \frac{1}{3}\sqrt{3}\right), \infty\right) \tag{2.4}$$

Tehát a derivált a $(-\infty, 1 - \frac{1\sqrt{3}}{3})$ és a $(1 + \frac{1\sqrt{3}}{3}, \infty)$ intervallumban szigorúan monoton növekvõ, másutt szigorúan monoton csökkenõ. Szemléltessük is mindezt:

```
> plot([f(x),D(f)(x)],x=-0.5..2.5,-2..3,color=[blue,red,black],
  thickness=[2,1],legend=["f" ,"D(f)"]);
```



>

A helyi szélsőérték elégséges feltételei

Tudjuk, hogy ha f deriválható, akkor csak olyan x_0 pontban lehet helyi szélsõértéke, ahol deriváltja nulla értékû, azaz $\left.\frac{d}{dx} f\right|_{x=x_0} = 0$

Azokat az x_0 pontokat, ahol a függvény elsõ deriváltja 0 értékû, azaz $\left.\frac{d}{dx} f\right|_{x=x_0} = 0$, a függvény stacionárius pontjainak mondjuk.A stacionárius pont azonban nem mindig szélsõértékhely! Jól mutatja ezt a következõ példa:

```
> f:=x->x^3;
```

$$f := x \rightarrow x^3 \qquad \textbf{(3.1)}$$

```
> plot([f(x),D(f)(x)],x=-1..1,color=[blue,red],title=`f = kék, f' = piros`);
```



>

Jóllehet az f deriváltja az $x_0 = 0$ helyen 0, a függvénynek nincs szélsõértéke. A helyi szélsõérték létezésére két elégséges föltételt is megfogalmazhatunk:

Első elégséges feltétel

A helyi szélsőérték létezésének első elégséges föltétele:

Tétel.

Ha az f függvény differenciálhányadosa az x_0 pontban 0, azaz $f'(x_0) = 0$, és f ' (x) az x_0 ponton való áthaladáskor előjelet vált, akkor az f függvénynek az x_0 pontban helyi szélsőértéke van.

Az előjelváltás azt jelenti, hogy van az x_0-nak olyan $(x_0 - \delta, x_0)$ és $(x_0, x_0 + \delta)$ bal- és jobboldali félkörnyezete, amelyekben a derivált eltérő előjelű.

Részletezve: Ha a derivált " pluszból minuszba vált":

x	$(x_0 - \delta, x_0)$	x_0	$(x_0, x_0 + \delta)$
f '(x)	+	0	-
$f(x)$	szigorúan monoton nő	helyi maximum	szigorúan monoton csökken

Ha pedig a derivált "minuszból pluszba vált":

x	$(x_0 - \delta, x_0)$	x_0	$(x_0, x_0 + \delta)$

f '(x)	-	0	+
$f(x)$	szigorúan monoton csökken	helyi minimum	szigorúan monoton nő

Második elégséges feltétel

Néha kényelmesebben kezelhető a következő - **második - elégséges föltétel:**

Tétel

Ha $\frac{d}{dx} f(x_0) = 0$ (másképp jelölve $f'(x_0) = 0$) és $\frac{d^2}{dx^2} f(x_0) \neq 0$ (másképp jelölve $f''(x_0) \neq 0$) , akkor az f függvénynek az x_0 helyen helyi szélsőértéke van. Mégpedig, ha $\frac{d^2}{dx^2} f(x_0) < 0$, akkor helyi maximuma, ha $0 < \frac{d^2}{dx^2} f(x_0)$, akkor helyi minimuma van az f függvénynek.

Példa

Vizsgáljuk meg az $f(x) = x^3 - 5x^2 - 4x + 1$ függvényt monotonitás és szélsõérték szempontjából!

Megoldás.

```
> restart:
> f:=x->x^3-5*x^2-4*x+1:
```

A függvény deriváltja:

```
> D(f)(x);
```

$$3x^2 - 10x - 4 \tag{3.3.1}$$

Megkeressük a derivált zérushelyeit:

```
> hely:=[solve(D(f)(x),x)];
```

$$hely := \left[\frac{5}{3} + \frac{1}{3}\sqrt{37}, \frac{5}{3} - \frac{1}{3}\sqrt{37}\right] \tag{3.3.2}$$

```
> zh:=evalf(hely);
```

$$zh := [3.694254177, -.360920843] \tag{3.3.3}$$

Növekvõleg rendezzük a zérushelyeket:

```
> zh:=sort(zh);
```

$$zh := [-.360920843, 3.694254177] \tag{3.3.4}$$

Ábrázoljuk a függvényt, az elsõ deriváltat és anak elõjelét a szélsõértékeket tartalmazó intervallumban:

```
> if nops(zh)=0 then print(`nincs kritikus pont`) else
> plot([f(x),D(f)(x),signum(D(f)(x))],x=zh[1]-1..zh[nops(zh)]
+1,color=[blue,red,green],thickness=[2,1,1],title=`f(x) =
kék, f' (x) = piros, f' elõjele = zöld`,discont=true,
```

```
legend=["f","f'","az f' elõjele"]):
> fi;
```



A derivált zérushelyeit, az ezeken a helyeken felvett függvényértékeket és a második derivált ezeken a helyeken felvett értékét tartalmató táblázat:

```
> if  nops(zh)<>0 then
> array([[`zérushely`,'f(`zérushely`)',(d^2*f/dx^2)],seq([zh[i],evalf(f(zh[i])),evalf((D@@2)(f)(zh[i]))],i=1..nops(zh))]) fi;
```

$$\begin{bmatrix} zérushely & f(zérushely) & \frac{d^2 f}{dx^2} \\ -.360920843 & 1.745349157 & -12.16552506 \\ 3.694254177 & -31.59720099 & 12.16552506 \end{bmatrix} \tag{3.3.5}$$

A szélsõértékhelyek a második elégséges feltétel alapján.

```
> if nops(zh)<>0 then for i to nops(zh) do if (D@@2)(f)(zh[i])>0 then print(array([zh[i],minhely])) elif (D@@2)(f)(zh[i])<0 then print(array([zh[i],maxhely])) elif (D@@2)(f)(zh[i])=0 then print(`igy nem dönthetünk`) ;fi od fi;
```

$$\begin{bmatrix} -.360920843 & maxhely \end{bmatrix}$$

$$\begin{bmatrix} 3.694254177 & minhely \end{bmatrix} \tag{3.3.6}$$

```
>
```

>

A szélsőérték részletező kiszámítása

Példa.

Határozzuk meg az

$$f(x) = x^2\, e^{(-x^2)}$$

függvény momotonitási intervallumait és helyi szélsőértékeit!

Megoldás.

A függvény páros, hiszen nyilvánvalóan $f(-x) = f(x)$ teljesül minden valós x értékre. Az f minden valós x értékre differenciálható.
Képezzük a f deriváltját és keressük meg a derivált zérushelyeit:

```
> f:=x->x^2*exp(-x^2);
```

$$f := x \to x^2\, e^{-x^2} \quad (4.1)$$

```
> Diff(f(x),x)=D(f)(x);
```

$$\frac{d}{dx}\left(x^2\, e^{-x^2}\right) = 2\, x\, e^{-x^2} - 2\, x^3\, e^{-x^2} \quad (4.2)$$

A deriváltat szorzattá alakítva a zérushelyek most közvetlenül is leolvashatók:

```
> Diff(f(x),x)=factor(D(f)(x));
```

$$\frac{d}{dx}\left(x^2\, e^{-x^2}\right) = -2\, x\, e^{-x^2}\, (x-1)\, (x+1) \quad (4.3)$$

Az eljárás továbbvitele érdekében oldjuk meg a $(D(f))(x) = 0$ egyenletet:

```
> zh:=[solve(D(f)(x))];
```

$$zh := [0, 1, -1] \quad (4.4)$$

A zérushelyek listáját alkottuk meg. A lista egyes elemeire hivatkozhatunk:

```
> zh[2];
```

$$1 \quad (4.5)$$

Rendezzük növekvõleg a derivált zérushelyeinek listáját:

```
> zh:=sort(zh);
```

$$zh := [-1, 0, 1] \quad (4.6)$$

A map utasítás segítségével egyszerre kiszámíthatjuk minden zérushelyen a második derivált értékét:

```
> d2zh:=map(x->'(D@@2)(f)'(x)=(D@@2)(f)(x),zh);
```

$$d2zh := \left[D^{(2)}(f)(-1) = -4\, e^{-1}, D^{(2)}(f)(0) = 2, D^{(2)}(f)(1) = -4\, e^{-1}\right] \quad (4.7)$$

A második derivált egyik helyen sem 0, tehát mindhárom helyen helyi szélsõértéke van az f függvénynek. Az elsõ és a harmadik helyen a második derivált negatív, ezért ezeken a helyeken az f-nek helyi maximuma van. Az $x = 0$ helyen pozitív a második derivált, ezért itt helyi minimuma van az f függvénynek.
Ezek értéke rendre:

```
> map(x->'f(x)'=f(x),zh);
```

(4.8)

$$\left[f(-1)=e^{-1}, f(0)=0, f(1)=e^{-1}\right] \tag{4.8}$$

Az alábbi utasítássorozattal is megkaphatjuk a fenti eredményeket:

```
> for i to nops(zh) do
>  if evalf((D@@2)(f)(zh[i]))<0 then
>    lprint(`Az`,x=zh[i],`pontban helyi maximum van, értéke=`,f(zh[i]))
>    elif evalf((D@@2)(f)(zh[i]))>0
>      then
>    lprint(`Az`,x=zh[i],`pontban helyi minimum van, értéke=`,f(zh[i]))
>    else lprint(`Az`,x=zh[i],`pontban így nem tudunk dönteni a helyi szélsõértékrõl`)
>  fi:
> od;
Az, x = -1, `pontban helyi maximum van, értéke=`, exp(-1)
Az, x = 0, `pontban helyi minimum van, értéke=`, 0
Az, x = 1, `pontban helyi maximum van, értéke=`, exp(-1)
```

Ezután ábrázoljuk a függvényt jellemzõ intervallumon:

```
> plot(f(x),x=zh[1]-2..zh[3]+2,color=blue,thickness=2);
```



Példa.

Határozzuk meg az

$$f(x)=\frac{x^5}{5}-\frac{3x^4}{4}-x^3+\frac{11x^2}{2}-6x$$

függvény momotonitási intervallumait és helyi szélsõértékeit!

Megoldás.

Az f minden valós x értékre differenciálható. Képezzük a f deriváltját és keressük meg a derivált zérushelyeit:

```
> f:=x->x^5/5-3/4*x^4-x^3+11/2*x^2-6*x;
```

$$f := x \rightarrow \frac{1}{5}x^5 - \frac{3}{4}x^4 - x^3 + \frac{11}{2}x^2 - 6x \tag{4.9}$$

```
> Diff(f(x),x)=D(f)(x);
```

$$\frac{d}{dx}\left(\frac{1}{5}x^5 - \frac{3}{4}x^4 - x^3 + \frac{11}{2}x^2 - 6x\right) = x^4 - 3x^3 - 3x^2 + 11x - 6 \tag{4.10}$$

A deriváltat szorzattá alakítva a zérushelyek most közvetlenül is leolvashatók:

```
> Diff(f(x),x)=factor(D(f)(x));
```

$$\frac{d}{dx}\left(\frac{1}{5}x^5 - \frac{3}{4}x^4 - x^3 + \frac{11}{2}x^2 - 6x\right) = (x+2)\,(x-3)\,(x-1)^2 \tag{4.11}$$

Az eljárás továbbvitele érdekében oldjuk meg a $(\mathrm{D}(f))(x) = 0$ egyenletet:

```
> zh:=[solve(D(f)(x))];
```

$$zh := [-2, 3, 1, 1] \tag{4.12}$$

Figyeljük meg, a derivált szorzat alakja az $(x-1)^2$ tényezõt tartalmazza. Ez azt jelenti, hogy az 1 kétszeres gyöke a deriváltnak. Ezért, bármelyik oldalról közelítünk az 1-hez, a derivált ugyanolyan előjelű értékeken át tart a 0-hoz, vagyis a differenciálhányados nem vált előjelet az $x = 1$ helyen. Tehát itt nem lesz szélsőérték! De végezzzük el részleteiben a vizsgálatot!

A zérushelyek listáját halmazzá alakítjuk, így a többszörös gyök csak egyszer jelenik meg. Ezután a halmazt ismét listává alakítjuk és növekvőleg rendezzük:

```
> zh:=convert(zh,set);
```

$$zh := \{-2, 1, 3\} \tag{4.13}$$

```
> zh:=sort(convert(zh,list));
```

$$zh := [-2, 1, 3] \tag{4.14}$$

A map utasítás segítségével egyszerre kiszámíthatjuk minden zérushelyen a második derivált értékét:

```
> d2zh:=map(x->(D@@2)(f)(x),zh);
```

$$d2zh := [-45, 0, 20] \tag{4.15}$$

A második derivált az elsõ és a harmadik helyen nem 0, ezeken a helyeken helyi szélsõértéke van az f függvénynek. Az elsõ helyen a deivált negatív, itt helyi maximum, a harmadik helyen pozitív, itt helyi minimum van.A második hely környezetében megvizsgáljuk a deriváltat

```
> plot(D(f)(x),x=zh[2]-0.5..zh[2]+0.5);
```



Az ábrán az látszik, hogy a derivált az $x = 1$ mindkét oldali félkörnyezetében negatív.Meg is kereshetjük azon x-ek halmazát, amelyekre $(D(f))(x) < 0$:

```
> solve(D(f)(x)<0);
```

$$RealRange(Open(-2), Open(1)), RealRange(Open(1), Open(3)) \tag{4.16}$$

Megállapíthatjuk, hogy az f függvénynek az $x = 1$ helyen nincs szélsõértéke, ezen a ponton csökkenõleg halad át.Ezután számítsuk ki az f függvény értékét a stacionárius pontokban:

```
> map(x->'f(x)'=f(x),zh);
```

$$\left[f(-2) = \frac{118}{5}, f(1) = -\frac{41}{20}, f(3) = -\frac{153}{20}\right] \tag{4.17}$$

Az alábbi utasítássorozattal is megkaphatjuk a fenti eredményeket:

```
> for i to nops(zh) do
>  if evalf((D@@2)(f)(zh[i]))<0 then
>     lprint(`Az`,x=zh[i],`pontban helyi maximum van, értéke=`,f
  (zh[i]))
>     elif evalf((D@@2)(f)(zh[i]))>0
>        then
>     lprint(`Az`,x=zh[i],`pontban helyi minimum van, értéke=`,f
  (zh[i]))
>     else lprint(`Az`,x=zh[i],`pontban így nem tudunk dönteni a
  helyi szélsõértékrõl`)
>  fi:
> od;
Az, x = -2, `pontban helyi maximum van, értéke=`, 118/5
Az, x = 1, `pontban így nem tudunk dönteni a helyi szélsõértékrõl`
```

```
Az, x = 3, `pontban helyi minimum van, értéke=`, -153/20
```

Ezután ábrázoljuk a függvényt jellemzõ intervallumon:

```
> plot(f(x),x=zh[1]-1..zh[nops(zh)]+1.1,color=blue,thickness=2)
;
```




```
>
```

Eljárások a helyi szélsőértékre

A solve használatával

A következő eljárások, közvetlenül, akkor használhatók, amikor a solve utasítással meg tudjuk keresni a derivált összes zérushelyét. Ha ez nem sikerül akkor is eredményesek lehetünk az evalf utasítás "bevetésével".

A **stacionarius** eljárás megkeresi a stcionárius helyeket.

```
> stacionarius:=proc()
   local hely,L,i:
    [solve(D(f)(x))]:
    hely:=map(allvalues,%):
    L:=NULL:
    for i to nops(hely) do
     if type(hely[i],realcons)
      then L:=L,hely[i]
     fi
    od:
    {L};
    sort(convert(%,list),(x,y)->is(x<y)):
```

```
end:
> f:=x->x^5-9*x^3;
```

$$f := x \rightarrow x^5 - 9\,x^3 \quad \textbf{(5.1.1)}$$

```
> helyek:=stacionarius(f);
```

$$helyek := \left[-\frac{3}{5}\sqrt{15},\, 0,\, \frac{3}{5}\sqrt{15} \right] \quad \textbf{(5.1.2)}$$

A **szelsoertek** eljárás a második elégséges feltétel alapján _"dolgozik", paramérerei az f fügvény és egy stacionárius hely. Csak akkor "mûködik", ha valóban stacionárius helyen hívjuk meg!

```
> szelsoert:=proc()
   local ert:
    ert:=evalf(subs(x=args[2],diff(args[1],x$2))):
     if ert<0 then lprint(`Az`,x=args[2],`pontban helyi
maximum van`)
      elif
     ert>0  then lprint(`Az`,x=args[2],`pontban helyi  minimum
van`)
     else lprint(`Az`,x=args[2],`Így nem tudunk dönteni`)
    fi:
end:
```

Nézzük például az elsõ stacionárius helyet:

```
> szelsoert(f(x),helyek[1]);
Az, x = -(3/5)*15^(1/2), `pontban helyi maximum van`
```

A szelsoert eljárást a helyek listájának minegyik elemére meghívjuk a for ciklussal:

```
> for i to nops(helyek) do
> szelsoert(f(x),helyek[i]);
> od;
Az, x = -(3/5)*15^(1/2), `pontban helyi maximum van`
Az, x = 0, `Így nem tudunk dönteni`
Az, x = (3/5)*15^(1/2), `pontban helyi minimum van`
> plot([f(x),D(f)(x)],x=min(op(helyek))-1..max(op(helyek))+1,
  -50..50,color=[blue,red],thickness=[2,1],legend=["f(x)","D
  (f)(x)"]);
```




```
>
```

Láthatóan az $x = 0$ helyen nincs szélsõérték!

Az fsolve használatával

Sok esetben a solve eljárás nem vezet eredményre. Máskor pedig adott intervallumon keresük a szélsõértékeket. Ilyenkor alkalmazhatjuk a következõ, **szelsoertek**, eljarast. Ennek paraméterei rendre az f függvény, az intervallum baloldali és jobboldali végpontja. Az eljárás csak akkor mûködik, ha az f folytonos az adott intervalumon.

```
> szelsoertek:=proc()
>  local  szakad,gyok,g,f,a,b,i:
>  global  ertekek:
>  f:=args[1]:
>  a:=args[2];
>  b:=args[3]:
> if not(iscont(f(x),x=a..b,'closed')) then RETURN(`Az f nem
  folytonos az adott intervallumon`) fi:
>  g:=x->1/D(f)(x);
>  szakad:=fdiscont(g(x),x=a..b,0.00001);
>  if nops(szakad)=0 then RETURN(`Az adott intervallumon
  nincs stacionárius hely`):
>  fi:
> for i to nops(szakad) do
  gyok[i]:=op({fsolve(D(f)(x),x=szakad[i])}) od;
```

```
> er t ekek: =[ a, seq( gyok[ i ] , i =1. . nops( szakad) ) , b] :
> f or  i  t o   nops( er t ekek) - 2  do
    i f  D( f ) ( ( er t ekek[ i ] +er t ekek[ i +1] ) / 2) <0  and  D( f ) ( ( er t ekek
  [ i +1] +er t ekek[ i +2] ) / 2) >0  t hen  l pr i nt ( ` Az` , x=gyok[ i ] ,
  ` pont ban  hel yi   mi ni mum van  ér t éke` , f ( gyok[ i ] ) )  f i :
>   i f  D( f ) ( ( er t ekek[ i ] +er t ekek[ i +1] ) / 2) >0  and  D( f ) ( ( er t ekek
  [ i +1] +er t ekek[ i +2] ) / 2) <0  t hen  l pr i nt ( ` Az` , x=gyok[ i ] ,
  ` pont ban  hel yi   maxi mum van  ér t éke` , f ( gyok[ i ] ) )  f i :
>   i f  D( f ) ( ( er t ekek[ i ] +er t ekek[ i +1] ) / 2) * D( f ) ( ( er t ekek[ i +1] +
  er t ekek[ i +2] ) / 2) >0  t hen  l pr i nt ( ` Az` , x=gyok[ i ] , ` pont ban
  ni ncs  hel yi   szél sõér t ék` )  f i :
> od;
> end:
>
>
```

A szelsoertek eljáráson alapul a **rajzolas** eljárás. Ez felhasználja a szelsoertek eljárás **ertekek** nevû globális paraméterét, amely a stacionárius pontok listája.
Paraméterei az f függvény, majd az ábrázolás határai az x ill. az y tengelyen.

```
> r aj zol as: =pr oc( )
> l ocal   f , er t ekek, a, b, pont ok,   pont ok1, pont ok2, vonal , i ,
  vonal ak, pont kep, kep, c, d:
> f : =ar gs[ 1] :
> er t ekek: =ar gs[ 2] :
> a: =ar gs[ 3] :
> b: =ar gs[ 4] :
> i f   not ( i scont ( f ( x) , x=a. . b, ' cl osed' ) )  t hen  RETURN( ` Az  f   nem
  f ol yt onos  az  adot t   i nt er val l umon` )  f i :
> c: =ar gs[ 5] : d: =ar gs[ 6] :
> pont ok1: =er t ekek[ 2. . nops( er t ekek) - 1] ;
> pont ok2: =map( x- >f ( x) , pont ok1) ;
> pont ok: =zi p( ( x, y) - >[ x, y] , pont ok1, pont ok2) ;
> f or  i  t o   nops( pont ok1)  do  vonal | | i : =pl ot t ool s[ l i ne] (
  [ pont ok1[ i ] , 0] , [ pont ok1[ i ] , f ( pont ok1[ i ] ) ] , col or =gr een,
  t hi ckness=2, l i nest yl e=3)
> od:
> vonal ak: =pl ot s[ di spl ay] ( { vonal | | ( 1. . nops( pont ok1) ) } ) :
> kep: =pl ot ( [ f ( x) , D( f ) ( x) ] , x=a. . b, c. . d, col or =[ bl ue, r ed] ,
  t hi ckness=[ 2, 2] ) :
> pont kep: =pl ot ( pont ok, st yl e=poi nt , symbol =ci r cl e, col or =r ed) :
> pl ot s[ di spl ay] ( { pont kep, kep, vonal ak} ) ;
> end:
```

Példa.

Határozzuk meg az

$$f(x) = e^{\left(-\frac{x}{10}\right)} \sin(x)$$

szélsõértékeit a $[-10, 10]$ *intervallumon!*

Megoldás.

```
> a: =- 10; b: =10;
```

$$a := -10$$

$$b := 10 \qquad \textbf{(5.2.1)}$$

```
> f : =x- >exp( - x/ 10) * si n( x) ;
```

$$f := x \to e^{-\frac{1}{10}x} \sin(x) \qquad \textbf{(5.2.2)}$$

Elõször a szelsoertek eljárást hívjuk meg.

```
> szel soer t ek( f , a, b) ;
Az, x = -7.953650286, `pontban helyi minimum van értéke`, -2.204255617
Az, x = -4.812057633, `pontban helyi maximum van értéke`, 1.609994235
Az, x = -1.670464979, `pontban helyi minimum van értéke`, -1.175944122
Az, x = 1.471127674, `pontban helyi maximum van értéke`, .8589127508
Az, x = 4.612720328, `pontban helyi minimum van értéke`, -.6273521845
Az, x = 7.754312981, `pontban helyi maximum van értéke`, .4582197239
```

A rajzolás eljárást ezután hívjuk meg. Az ábrázolás paramétereivel természetesen kísérleteznünk kell, ha valóban "szép" ábrát akarunk kapni.

```
> r aj zol as( f , er t ekek, a+1, b- 1, - 2. 5, 2) ;
```



Példa.

Vizsgáljuk meg az

$$f(x) = e^x - x^3 - 2$$

függvény szélsõértékeit!

Megoldás.

```
> f:=x->exp(x)-x^3-2;
```

$$f := x \rightarrow e^x - x^3 - 2 \tag{5.2.3}$$

Némi kísérletezés után rábukkanhatunk arra az intervallumra, amelyen szélsõértéket nyerhetünk.

```
> plot(f(x),x=-3..5,color=blue,thickness=2);
```



Az intervallum végpontjai legyenek:

```
> a:=-3:b:=5:
```

```
> solve(D(f)(x));
```

$$-2\,\mathrm{LambertW}\left(-\frac{1}{6}\sqrt{3}\right), -2\,\mathrm{LambertW}\left(-1, -\frac{1}{6}\sqrt{3}\right), -2\,\mathrm{LambertW}\left(\frac{1}{6}\sqrt{3}\right) \tag{5.2.4}$$

A solve közvetlenül nem adja meg a derivált zérushelyeit. Dolgozzunk a szelsoertek eljárással!

```
> szelsoertek(f,a,b);
Az, x = -.4589622675, `pontban helyi minimum van értéke`, -1.271382178
Az, x = .9100075725, `pontban helyi maximum van értéke`, -.269248466
Az, x = 3.733079029, `pontban helyi minimum van értéke`, -12.21610066
```

```
> rajzolas(f,ertekek,a,b,-13,11);
```



Természetesen használhatjuk a stacionarius és a szelsoert eljárásokat is. Ekkor azonban az evalf utasítás közbevetésével kell a stacionárius helyek értékét "kikényszeríteni:

```
> helyek:=stacionarius(f);
```

$$helyek := \left[-2\,\mathrm{LambertW}\left(\frac{1}{6}\sqrt{3}\right),\ -2\,\mathrm{LambertW}\left(-\frac{1}{6}\sqrt{3}\right),\ -2\,\mathrm{LambertW}\left(-1,\ -\frac{1}{6}\sqrt{3}\right)\right] \tag{5.2.5}$$

```
> helyek:=evalf(helyek);
```

$$helyek := [-.4589622676,\ 0.9100075724,\ 3.733079028] \tag{5.2.6}$$

```
> for i to nops(helyek) do
> szelsoert(f(x),helyek[i]);
> od;
Az, x = -.4589622676, `pontban helyi minimum van`
Az, x = .9100075724, `pontban helyi maximum van`
Az, x = 3.733079028, `pontban helyi minimum van`
```

Példa.

Vizsgáljuk meg szélsõérték szempontjából az

$$f(x) = x^3 - x^5$$

függvényt a $[-1, 1]$ *intervallumon!*

```
> f:=x->x^3-x^5;
```

$$f := x \to x^3 - x^5 \tag{5.2.7}$$

```
> szelsoertek(f,-1,1);
Az, x = -.7745966692, `pontban helyi minimum van értéke`, -.1859032006
Az, x = 0., `pontban nincs helyi szélsõérték`
```

```
Az, x = .7745966692, `pontban helyi maximum van értéke`, .1859032006
```

```
> rajzolas(f,ertekek,-1.3,1.3,-2,2);
```




```
>
```

Példák

Példa.

Vizsgáljuk meg monotonitás és szélsõérték szempontjából az

$$f(x) = x^k e^x$$

alakú függvényeket, ahol a k pozitív egész szám!

Megoldás.

vizsgáljuk elõször a $k = 1$ esetet!

```
> restart:
> f:=x->x*exp(x);
```

$$f := x \to x\, e^x \tag{6.1}$$

A függvény minden valós x-re differenciálható. Keressük meg a derivált zérushelyeit:

```
> Diff(f(x),x)=diff(f(x),x);
```

$$\frac{d}{dx}\left(x\, e^x\right) = e^x + x\, e^x \tag{6.2}$$

Szorzattá alakítva:

```
> Diff(f(x),x)=factor(diff(f(x),x));
```

$$\frac{d}{dx}\left(x\, e^x\right) = e^x\,(1 + x) \tag{6.3}$$

```
> zh:=solve(rhs(%),x);
```

$$zh := -1 \tag{6.4}$$

Ezen a helyen a második derivált:

```
> '(D@@2)'(f)(zh)=(D@@2)(f)(zh);
```

$$D^{(2)}(f)(-1) = e^{-1} \tag{6.5}$$

Ez pozitív, tehát a függvénynek itt helyi minimuma van. Ebbõl az is következik, hogy ennél kisebb x értékekre szigorúan monoton csökkenõ, ennél nagyobb x értékekre pedig monoton növekvõ a függvény.

```
> plot(f(x),x=-8..0.4);
```



Legyen most $k = 2$!

```
> f:=x->x^2*exp(x);
```

$$f := x \to x^2 e^x \tag{6.6}$$

Ez a függvény is minden valós x-re differenciálható. Keressük meg a derivált zérushelyeit:

```
> Diff(f(x),x)=diff(f(x),x);
```

$$\frac{d}{dx}\left(x^2 e^x\right) = 2\,x\,e^x + x^2\,e^x \tag{6.7}$$

Szorzattá alakítva:

```
> Diff(f(x),x)=factor(diff(f(x),x));
```

$$\frac{d}{dx}\left(x^2 e^x\right) = x\,e^x\,(2 + x) \tag{6.8}$$

```
> zh:=[solve(rhs(%),x)];
```

$$zh := [0, -2] \tag{6.9}$$

Ezeken a helyen a második derivált:

```
> '(D@@2)(f)'(zh[1])=(D@@2)(f)(zh[1]);
```

$$D^{(2)}(f)(0) = 2 \tag{6.10}$$

Tehát az $x = 0$ helyen helyi (és abszolút) minimuma van a függvénynek.

```
> '(D@@2)(f)'(zh[2])=(D@@2)(f)(zh[2]);
```

$$D^{(2)}(f)(-2) = -2\,e^{-2} \qquad (6.11)$$

Tehát az $x = -2$ helyen helyi maximuma van a függvénynek. A monotonitási viszonyok ebbõl már adódnak, s az ábráról is látszanak:

```
> plot(f(x),x=-9..0.55);
```



A függvény a -2-nél kisebb x értékekre szigorúan monoton növekvõ, a (-2,0) intervallumban szigorúan monoton csökkenõ, pozitív x-ekre szigorúan monoton növekvõ.
Vizsgáljuk meg most a függvényt általában:

```
> f:=x->x^k*exp(x);
```

$$f := x \rightarrow x^k e^x \qquad (6.12)$$

A derivált:

```
> (D(f)(x));
```

$$\frac{x^k k\, e^x}{x} + x^k e^x \qquad (6.13)$$

Hozzuk közös nevezõre:

```
> normal(%);
```

$$\frac{x^k e^x (k + x)}{x} \qquad (6.14)$$

Hozzuk egyszerûbb alakra:

```
> simplify(%);
```

$$x^{k-1} e^x (k + x) \qquad (6.15)$$

A Maple bonyolult derivált-kezelése ne tévesszen meg bennünket, a függvény minden x-re deriválható. Az is látszik, hogy a derivált az $x = -k$ és az $x = 0$ helyeken 0 értékû. Az $x = -k$ helyen mindig elõjelet vált a függvény, páratlan k értékek esetén negatívból pozitívba, páros k értékek esetén pedig pozitívból negatívba vált. Tehát az $x = -k$ helyen minden pozitív egész k esetén helyi szélsõértéke van a függvénynek, páratlan k esetén minimuma, páros k esetén maximuma. Az $x = 0$ helyen páros k értékek esetén van szélsõérték és ez minimum.
Az elsõ hat k értékre adódó függvények képe látható a következõkben:

```
> for k to 6 do
>   kep||k:=plot(x^k*exp(x),x=-k-5..0.3*k,color=blue,thickness=2):
> od:
> plots[display](array(1..2,1..3,[[seq(kep||k,k=1..3)],[seq(kep||k,k=4..6)]]));
```



0,4
0,3
0,2
0,1
-6 -5 -4 -3 -2 -1 0
x
-0,1
-0,2
-0,3
0,6
0,5
0,4
0,3
0,2
0,1
-7 -6 -5 -4 -3 -2 -1 0
x
1,5
1,0
0,5
-8 -6 -4 -2 0
x
-0,5
-1,0
6
5
4
3
30
20
10
200
150
100

Példa.

Tekintsük az

$$f(x) = x \ln(x)^k$$

alakú függvényeket,ahol a k pozitív egész szám. Vizsgáljuk meg az ilyen alakú f függvényeket szélsőérték szempontjából!

Megoldás.

Elõször vizsgáljuk meg az első néhány k értékre, mondjuk $k = 1, 2, 3, 4, 5$ esetén az adódó függvényeket!
Tehát írjuk a k helyére rendre a fenti értékeket, majd "szabadítsuk fel a k értékét a k := 'k' utasítással!
Mi csak egy konkrét példát és az általános esetet "futtatjuk" le.
a) **Speciális értékek esetén.**
Tehát itt javasoljuk, hogy írjon a k helyére a 2 helyett rendre 1, 3, 4, 5 értéket és figyelje az adódó eredményeket!

```
> f:=x->x*ln(x)^k; k:=2:
```

$$f := x \to x \ln(x)^k \qquad (6.16)$$

A derivált, szorzattá alkítva:

```
> df:=factor(D(f)(x));
```

$$df := \ln(x)\,(\ln(x) + 2) \qquad (6.17)$$

Hozzuk a deriváltat egyszerûbb alakra, és készítsünk belõle függvényt:

```
> df:=unapply(simplify(%),x);
```

$$df := x \to \ln(x)\,(\ln(x) + 2) \qquad (6.18)$$

Keressük meg és rendezzük a derivált zérushelyeit:

```
> zh:=sort([solve(df(x),x)],(x,y)->is(x<y));
```

$$zh := \left[\frac{1}{e^2}, 1\right] \qquad (6.19)$$

Láthatóan páros k esetén a derivált az 1 és az $\frac{1}{e^k}$ is helyen előjelet vált, páratlan k esetén pedig csak az $\frac{1}{e^k}$ helyen, Az 1 helyen az elõjelváltás negatívból pozitívba, az $\frac{1}{e^k}$ helyen pozitívból negatívba történik !
A fenti eredményeket a következõ utasítássorozattal is megkaphatjuk:

```
> hatarok:={0,4};
```

$$hatarok := \{0, 4\} \qquad (6.20)$$

```
> ertekek:=convert(zh,set) union hatarok;
```

$$ertekek := \left\{0, 1, 4, \frac{1}{e^2}\right\} \qquad (6.21)$$

```
> ertekek:=sort(convert(ertekek,list),(x,y)->is(x<y));
```

$$ertekek := \left[0, \frac{1}{e^2}, 1, 4\right] \qquad (6.22)$$

```
>
> for i to nops(ertekek)-2 do
```

```
if evalf(D(f)((ertekek[i]+ertekek[i+1])/2))<0 and evalf(D(f)
((ertekek[i+1]+ertekek[i+2])/2))>0 then lprint(`Az`,x=ertekek
[i+1],`pontban helyi minimum van értéke`,f(ertekek[i+1])) fi:
> if evalf(D(f)((ertekek[i]+ertekek[i+1])/2))>0 and evalf(D(f)
((ertekek[i+1]+ertekek[i+2])/2))<0 then lprint(`Az`,x=ertekek
[i+1],`pontban helyi maximum van értéke`,f(ertekek[i+1])) fi:
> if evalf(D(f)((ertekek[i]+ertekek[i+1])/2)*D(f)((ertekek
[i+1]+ertekek[i+2])/2))>0 then lprint(`Az`,x=ertekek[i+1],
`pontban nincs helyi szélsõérték`) fi:
> od;
Az, x = 1/exp(2), `pontban helyi maximum van értéke`, 4/exp(2)
Az, x = 1, `pontban helyi minimum van értéke`, 0
```

```
> plot(f(x),x=0..1.8,color=blue,thickness=2);
```



b) **Most jöjjön az általánosítás!**

```
> f:=x->x*ln(x)^k; k:='k':
```

$$f := x \to x \ln(x)^k \tag{6.23}$$

A derivált, szorzattá alkítva:

```
> df:=factor(D(f)(x));
```

$$df := \frac{\ln(x)^k\,(\ln(x) + k)}{\ln(x)} \tag{6.24}$$

Hozzuk a deriváltat egyszerűbb alakra, és készítsünk belőle függvényt:

```
> df:=unapply(simplify(%),x);
```

$$df := x \to \ln(x)^{k-1}\,(\ln(x) + k) \tag{6.25}$$

Keressük meg és rendezzük a derivált zérushelyeit:

```
> zh:=sort([solve(df(x),x)],(x,y)->is(x<y));
```

(6.26)

$$zh := \left[\frac{1}{e^k}, 1\right] \quad \mathbf{(6.26)}$$

Láthatóan páros k esetén a derivált az 1 és az $\frac{1}{e^k}$ is helyen elõjelet vált. Az 1 helyen az elõjelváltás negatívból pozitívba, az $\frac{1}{e^k}$ helyen pozitívból negatívba történik. Páratlan k esetén pedig csak az $\frac{1}{e^k}$ helyen vált a derivált elõjelelet és negatívból pozitívba. Általánosan is igazolt eredményünk tehát :
Ha a k páratlan, akkor csak az $x = \frac{1}{e^k}$ helyen van szélsõérték és ez minimum, ha a k páros, akkor az $x = \frac{1}{e^k}$ helyen helyi maximum van, at $x = 1$ helyen pedig helyi minimum. A szélsõértékek értéke pedig:

```
> map(x->'f(x)'=simplify(f(x),symbolic),zh);
```

$$\left[f\left(\frac{1}{e^k}\right) = (-1)^k k^k e^{-k}, f(1) = 0\right] \quad \mathbf{(6.27)}$$

```
>
```

Példa.

Határozzuk meg, hogy az a valós paramétertől függően hány valós gyöke van az

$$x + \ln\left(\left(\frac{x+1}{x}\right)^2\right) = a$$

egyenletnek!

Megoldás.

Az egyenlet bal oldalát alkotó kifejezésből képzett függvényt vizsgáljuk.Ennek értékkészlete és monotonitása határozza meg a gyökök számát.

```
> restart:
> f:=x->x+ln(((x+1)/x)^2);
```

$$f := x \to x + \ln\left(\frac{(x+1)^2}{x^2}\right) \quad \mathbf{(6.28)}$$

A függvény minden valós x-re értelmezhető, kivéve az $x = -1$ és az $x = 0$ értéket.

```
> plot(f(x),x=-3..3,color=blue,thickness=2);
```



Az ábra alapján úgy tűnik, hogy minden valós a mellett van gyöke az egyenletnek. A helyi maximumnál nagyobb ill. a helyi minimumnál kisebb a értékekre 3 gyök van, a helyi szélsőértékekkel egyező értékek esetén 2 gyök van, egyébként 1 gyök adódik. Mindezeket igazoljuk a következőkben.

A két szakadási hely környezetében jellemezzük a függvényt először:

```
> Limit(f(x),x=-1)=limit(f(x),x=-1);
```

$$\lim_{x \to -1} \left(x + \ln\left(\frac{(x+1)^2}{x^2} \right) \right) = -\infty \tag{6.29}$$

```
> Limit(f(x),x=0)=limit(f(x),x=0);
```

$$\lim_{x \to 0} \left(x + \ln\left(\frac{(x+1)^2}{x^2} \right) \right) = \infty \tag{6.30}$$

Másutt a függvény mindenütt folytonos, sõt deriválható, mert a polinomfüggvények mindenütt deriválhatók, a logaritmus függvény minden pozitív változóértékre deriválható és a racionális törtfüggvény is mindenütt deriválható, kivéve a nevező zérushelyeit.
Képezzük a deriváltfüggvényt és határozzuk meg annak zérushelyeit, a lehetséges helyi szélsőértékhelyeket!

```
> Diff(f(x),x)=diff(f(x),x);
```

$$\frac{d}{dx} \left(x + \ln\left(\frac{(x+1)^2}{x^2} \right) \right) = 1 + \frac{\left(\frac{2\,(x+1)}{x^2} - \frac{2\,(x+1)^2}{x^3} \right) x^2}{(x+1)^2} \tag{6.31}$$

A deriváltat egyszerûbb alakra hozva:

```
> Diff(f(x),x)=simplify(diff(f(x),x));
```

$$\frac{d}{dx}\left(x+\ln\left(\frac{(x+1)^2}{x^2}\right)\right)=\frac{x^2+x-2}{x\,(x+1)} \qquad (6.32)$$

A derivált zérushelyei:

```
> zh:=[solve(rhs(%),x)];
```

$$zh := [1, -2] \qquad (6.33)$$

A második derivált értéke ezeken a helyeken:

```
> '(D@@2)(f)'(zh[1])=(D@@2)(f)(zh[1]);'(D@@2)(f)'(zh[2])=(D@@2)
  (f)(zh[2]);
```

$$D^{(2)}(f)(1)=\frac{3}{2}$$

$$D^{(2)}(f)(-2)=-\frac{3}{2} \qquad (6.34)$$

A map utasítás segítsével rövidebben is eljárhatunk:

```
> map(x->'(D@@2)(f)(x)'=(D@@2)(f)(x),zh);
```

$$\left[D^{(2)}(f)(1)=\frac{3}{2},\ D^{(2)}(f)(-2)=-\frac{3}{2}\right] \qquad (6.35)$$

Tehát az $x=-2$ pontban helyi maximum, az $x=1$ pontban helyi minimum van. Ezek értéke:

```
> 'f'(zh[1])=f(zh[1]);
```

$$f(1)=1+2\ln(2) \qquad (6.36)$$

```
> 'f'(zh[2])=f(zh[2]);
```

$$f(-2)=-2-2\ln(2) \qquad (6.37)$$

Ábrázoljuk a függvényt és első deriváltját közös rendszerben:

```
> plot([f(x),D(f)(x)],x=-8..8,-20..20,thickness=[2,2],color=
  [blue,red],discont=true);
```



A monotonitási viszonyok egyértelműen adódnak most már. Még az értékkészlet végleges megállapításához szükséges vizsgálat, a $-\infty$-ben és a ∞-ben vett határértékek:

```
> Limit(f(x),x=infinity)=limit(f(x),x=infinity);
```

$$\lim_{x\to\infty}\left(x+\ln\left(\frac{(x+1)^2}{x^2}\right)\right)=\infty \tag{6.38}$$

```
> Limit(f(x),x=-infinity)=limit(f(x),x=-infinity);
```

$$\lim_{x\to-\infty}\left(x+\ln\left(\frac{(x+1)^2}{x^2}\right)\right)=-\infty \tag{6.39}$$

```
> a1:=f(zh[1]);a2:=f(zh[2]);
```

$$a1:=1+2\ln(2)$$

$$a2:=-2-2\ln(2) \tag{6.40}$$

Igazoltuk tehát, hogy ha $a<-2-\ln(4)$ vagy $1+\ln(4)<a$, akkor 3 gyök van, ha a egyenlõ ezen értékek valamelyikével, akkor két gyök van, ha pedig a a két érték között van, akkor egy gyök van.

Úgy tűnik, hogy egyenes a függvény aszimptotája. Mikor is mondjuk,hogy az $y=m\,x+b$ egyenes aszimptotája az f függvénynek?

Definíció

Az $y=m\,x+b$ alakú egyenes az f függvény aszimptota egyenese, ha:

$$\lim_{x\to\infty}\left(\frac{f(x)}{x}\right)=m \quad \text{és} \quad \lim_{x\to\infty}(f(x)-m\,x)=b$$

teljesül.

Az m értékét tehát a következõ határérték adja:

```
> Limit(f(x)/x,x=infinity)=limit(f(x)/x,x=infinity);
```

$$\lim_{x\to\infty}\left(\frac{x+\ln\left(\frac{(x+1)^2}{x^2}\right)}{x}\right)=1 \tag{6.41}$$

```
> m:=rhs(%);
```

$$m:=1 \tag{6.42}$$

A b értékét az alábbi határérték szolgáltatja:

```
> Limit(f(x)-m*x,x=infinity)=limit(f(x)-m*x,x=infinity);
```

$$\lim_{x\to\infty}\ln\left(\frac{(x+1)^2}{x^2}\right)=0 \tag{6.43}$$

```
> b:=rhs(%);
```

$$b:=0 \tag{6.44}$$

Az **aszimptota:**

```
> y=m*x+b;
```

$$y=x \tag{6.45}$$

```
> aszimptota:=unapply(rhs(%),x);
```

$$aszimptota:=x\to x \tag{6.46}$$

```
>
```

```
> plot([f(x),aszimptota(x)],x=-6..6,title=`A függvény és aszimptotája`,color=[blue,red],thickness=[2,1]);
```



Az ábra alapján úgy tűnik, hogy f középpontosan szimmetrikus. A szimmetriaközéppont az aszimptota és a függvénygörbe metszéspontja lehet.
Határozzuk meg ezt a pontot:

```
> x0:=solve(f(x)=x);
```

$$x0 := -\frac{1}{2} \tag{6.47}$$

```
> f(x0);
```

$$-\frac{1}{2} \tag{6.48}$$

A szimmetriaközéppont tehát a $P_0\left(-\frac{1}{2}, -\frac{1}{2}\right)$ pont lehet. Toljuk el az f függvényt a $v = \left[\frac{1}{2}, \frac{1}{2}\right]$ vektorral:

```
> x:='x':
> g:=unapply(f(x-1/2)+1/2,x);
```

$$g := x \to x + \ln\left(\frac{\left(x+\frac{1}{2}\right)^2}{\left(x-\frac{1}{2}\right)^2}\right) \tag{6.49}$$

```
> plot([g(x),x],x=-10..10,-10..10,discont=true,color=[blue,red],thickness=[2,1]);
```



Az ábra azt mutatja, hogy a transzformált függvény páratlan. Igazoljuk ezt! A rendszerrel közölni kell, hogy x valós és $|x| \neq \frac{1}{2}$:

```
> assume(x,real); additionally(abs(x)<>1/2);
```

Azt kell megmutatni, hogy minden $|x| \neq \frac{1}{2}$ esetén $g(-x) = -g(x)$, azaz $g(x) + g(-x) = 0$

```
> g(x)+g(-x);
```

$$\ln\left(\frac{\left(x\sim+\frac{1}{2}\right)^2}{\left(x\sim-\frac{1}{2}\right)^2}\right) + \ln\left(\frac{\left(-x\sim+\frac{1}{2}\right)^2}{\left(-x\sim-\frac{1}{2}\right)^2}\right) \qquad \textbf{(6.50)}$$

Alkalmazzuk a logaritmusra vonatkozó azonosságot:

```
> combine(g(x)+g(-x));
```

$$\ln\left(\frac{\left(x\sim+\frac{1}{2}\right)^2\left(-x\sim+\frac{1}{2}\right)^2}{\left(x\sim-\frac{1}{2}\right)^2\left(-x\sim-\frac{1}{2}\right)^2}\right) \qquad \textbf{(6.51)}$$

Már az előbb is felfedezhető volt, de most már méginkább látszik állításunk igazsága. A simplify utasítás végrehajtása után ezt a rendszer is "felismeri".

```
> simplify(combine(g(x)+g(-x)));
```

$$0 \qquad \textbf{(6.52)}$$

Másképpen is eljárhatunk. Ismert ugyanis a következő:

Tétel.

Legyen az $f(x)$ függvény az I intervallumon (az I lehet zárt, nyitott, félig nyitott, véges vagy végtelen) értelmezett, folytonos és az intervallum belsejében differenciálható függvény. Ekkor az I intervallumon akkor és csak akkor állandó, ha f ' (x) = 0 teljesül az I intervallum minden belső pontjában.

A tétel szerint elég megmutatni, hogy
• a differenciálhányados azonosan 0 és
• emelllett egy pontban azt, hogy ott 0 a függvényérték:

```
> h:=unapply(g(x)+g(-x),x);
```

$$h := x\sim \rightarrow \ln\left(\frac{\left(x\sim+\frac{1}{2}\right)^2}{\left(x\sim-\frac{1}{2}\right)^2}\right) + \ln\left(\frac{\left(-x\sim+\frac{1}{2}\right)^2}{\left(-x\sim-\frac{1}{2}\right)^2}\right) \tag{6.53}$$

```
> Diff(h(x),x)=D(h)(x);
```

$$\frac{d}{dx\sim}\left(\ln\left(\frac{\left(x\sim+\frac{1}{2}\right)^2}{\left(x\sim-\frac{1}{2}\right)^2}\right) + \ln\left(\frac{\left(-x\sim+\frac{1}{2}\right)^2}{\left(-x\sim-\frac{1}{2}\right)^2}\right)\right) \tag{6.54}$$

$$= \frac{\left(\frac{2\left(x\sim+\frac{1}{2}\right)}{\left(x\sim-\frac{1}{2}\right)^2} - \frac{2\left(x\sim+\frac{1}{2}\right)^2}{\left(x\sim-\frac{1}{2}\right)^3}\right)\left(x\sim-\frac{1}{2}\right)^2}{\left(x\sim+\frac{1}{2}\right)^2}$$

$$+ \frac{\left(-\frac{2\left(-x\sim+\frac{1}{2}\right)}{\left(-x\sim-\frac{1}{2}\right)^2} + \frac{2\left(-x\sim+\frac{1}{2}\right)^2}{\left(-x\sim-\frac{1}{2}\right)^3}\right)\left(-x\sim-\frac{1}{2}\right)^2}{\left(-x\sim+\frac{1}{2}\right)^2}$$

Hát, erről aztán egyáltalálán nem látszik, hogy azonosan 0 értékű! De, szerencsére segít, a simplify utasítás, amely -jó esetben-a "végletekig" egyszerűsíti a függvényt defininiáló kifejezést:

```
> simplify(D(h)(x));
```

$$0 \tag{6.55}$$

Számítsunk ki, mondjuk az $x = 3$ helyen a függvény értékét:

```
> h(3);
```

$$\ln\left(\frac{49}{25}\right) + \ln\left(\frac{25}{49}\right) \tag{6.56}$$

```
> simplify(h(3));
```

$$0 \tag{6.57}$$

Tehát a derivált azonosan 0, így h állandó, de pl. $h(3) = 0$ miatt $g(x) + g(-x) = 0$, azaz az eltolt függvény páratlan.

Kérdések, feladatok

Ellenőrző kérdések

1.Milyen elégséges feltételt ismer arra vonatkozólag, hogy a differenciáható függvény az x_0 ponton szigorúan monoton növekvőleg haladjon át?
2.Differenciálható függvénynek mely pontokban lehet helyi szélsőértéke?
3. Hogyan szól Rolle tétele?
4. Mi a Lagrange-tétel geometriai jelentése?
5. Mi a feltétele annak, hogy az (a,b) intervallumon differenciálható függvény ezen az intervallumon csökkenő legyen?
6. Milyen elégséges feltételeit ismeri annak, hogy a differenciálható f függvénynek az x_0 helyen helyi szélsõértéke legyen?

Gyakorló feladatok

1. Vizsgáljuk meg, hogy az f függvény a megadott [a,b] intervallumban teljesíti-e a Lagrange-tétel feltételeit, s ha igen, adja meg az (a,b) intervallumban az összes olyan ξ értéket, amelyre a tétel teljesül.

1.1. $f(x)=x^3+1$, [-2,4]; 1.4. $f(x)=x^3-2x^2+x+3$, [-1,1];
1.2. $f(x)=x+\dfrac{4}{x}$, [1, 4]; 1.5. $f(x)=\dfrac{1}{(x-1)^2}$, [0,2];
1.3. $f(x)=4+\sqrt{x-1}$, [1,5]; 1.6. $f(x)=|x-3|$, [-1,4]

2. Legyen $P_1(x_1,y_1)$ és $P_2(x_2,y_2)$ az $y=a\,x^2+b\,x+c$ parabola két tetszõleges pontja, $P_3(x_3,y_3)$ a P_1P_2 ív olyan pontja amelyben a görbéhez húzott érintõ párhuzamos a P_1P_2 húrral. Bizonyítsuk be, hogy $x_3=\dfrac{x_1+x_2}{2}$.
3. Bizonyítsa be, felhasználva Rolle tételét, hogy egy harmadfokú polinomfüggvénynek legfeljebb három valós zérushelye lehet!
4. Határozza meg az alábbi függvények monotonitási intervallumait és helyi szélsőértékeit:
4.1. $f(x)=3x-x^3$; 4.2. $f(x)=x^3-6x^2+9x-4$; 4.3. $f(x)=(x-2)^5\,(2x+1)^4$; 4.4. $f(x)=\dfrac{\ln(x)^2}{x}$;
4.5. $f(x)=x^2-\ln\left(x^2\right)$; 4.6. $f(x)=2\sin(x)+\cos(2x)$; 4.7. $f(x)=\dfrac{4x}{x^2+1}$; 4.8. $f(x)=x\ln(x)^2$.